Derivazione errore per il metodo dei "tre ottavi"

```
> restart:
```

Integrale esatto su intervallo [-h,h]

```
> Iesatto := int( f(x),x=-3/2*h..3/2*h) ;
```

$$Iesatto := \int_{-\frac{3}{2}h}^{\frac{3}{2}h} f(x)\,dx \tag{1}$$

Integrazione approssimata con 1 intervallo di ampiezza 3*h col metodo 3/8

```
> I3_8 := (3*h/8)*(f(-(3/2)*h)+3*f(-h/2)+3*f(h/2)+f(3/2*h)) ;
```

$$I3\_8 := \frac{3}{8}h\left(f\left(-\frac{3}{2}h\right)+3f\left(-\frac{1}{2}h\right)+3f\left(\frac{1}{2}h\right)+f\left(\frac{3}{2}h\right)\right) \tag{2}$$

```
> Errore := Iesatto - I3_8 ;
```

$$Errore := \int_{-\frac{3}{2}h}^{\frac{3}{2}h} f(x)\,dx - \frac{3}{8}h\left(f\left(-\frac{3}{2}h\right)+3f\left(-\frac{1}{2}h\right)+3f\left(\frac{1}{2}h\right)+f\left(\frac{3}{2}h\right)\right) \tag{3}$$

Introduciamo funzione "trucco"

```
> G := int( f(x),x=-3/2*t..3/2*t)
       - (3*t/8)*(f(-3/2*t)+3*f(-t/2)+3*f(t/2)+f(3/2*t)) - E*(t/h)^N;
```

$$G := \int_{-\frac{3}{2}t}^{\frac{3}{2}t} f(x)\,dx - \frac{3}{8}t\left(f\left(-\frac{3}{2}t\right)+3f\left(-\frac{1}{2}t\right)+3f\left(\frac{1}{2}t\right)+f\left(\frac{3}{2}t\right)\right) - E\left(\frac{t}{h}\right)^N \tag{4}$$

"E" è l'errore (differenza tra integrale esatto e approssimato) del metodo dei "tre-ottavi".
Questa espressione va a zero quando t=h, infatti:

```
> subs(t=h,G) ;
```

$$\int_{-\frac{3}{2}h}^{\frac{3}{2}h} f(x)\,dx - \frac{3}{8}h\left(f\left(-\frac{3}{2}h\right)+3f\left(-\frac{1}{2}h\right)+3f\left(\frac{1}{2}h\right)+f\left(\frac{3}{2}h\right)\right) - E \tag{5}$$

"E" è l'errore, cioè quel numero che manda a zero la precedente espressione.
Quando t=0 abbiamo:

```
> eval(subs(t=0,G));
```

$$0 \tag{6}$$

Quindi la funzione G si azzera in 0 ed h, esiste punto 0<alpha<h tale che G'(alpha) = 0

```
> DG := simplify(diff(G,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DG := \frac{9}{8}f\left(\frac{3}{2}t\right)+\frac{9}{8}f\left(-\frac{3}{2}t\right)-\frac{9}{8}f\left(-\frac{1}{2}t\right)-\frac{9}{8}f\left(\frac{1}{2}t\right)+\frac{9}{16}t\,\mathrm{D}(f)\left(-\frac{3}{2}t\right) + \frac{9}{16}t\,\mathrm{D}(f)\left(-\frac{1}{2}t\right)-\frac{9}{16}t\,\mathrm{D}(f)\left(\frac{1}{2}t\right)-\frac{9}{16}t\,\mathrm{D}(f)\left(\frac{3}{2}t\right)-E\,t^{N-1}h^{-N}N \tag{7}$$

Valutando la funzione DG in 0

```
> subs(t=0,DG) ;
```

$$0 \quad \mathbf{(8)}$$

Quindi la funzione G' si azzera in 0 ed alpha, esiste punto 0<beta<alpha<h tale che G''(beta) = 0.

```
> DDG := simplify(diff(DG,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DDG := \frac{9}{8}\,\mathrm{D}(f)\left(\frac{3}{2}\,t\right) - \frac{9}{8}\,\mathrm{D}(f)\left(-\frac{3}{2}\,t\right) + \frac{9}{8}\,\mathrm{D}(f)\left(-\frac{1}{2}\,t\right) - \frac{9}{8}\,\mathrm{D}(f)\left(\frac{1}{2}\,t\right) - \frac{27}{32}\,t\,\mathrm{D}^{(2)}(f)\left(-\frac{3}{2}\,t\right) - \frac{9}{32}\,t\,\mathrm{D}^{(2)}(f)\left(-\frac{1}{2}\,t\right) - \frac{9}{32}\,t\,\mathrm{D}^{(2)}(f)\left(\frac{1}{2}\,t\right) - \frac{27}{32}\,t\,\mathrm{D}^{(2)}(f)\left(\frac{3}{2}\,t\right) - E\,t^{N-2}\,h^{-N}\,N^2 + E\,t^{N-2}\,h^{-N}\,N \quad \mathbf{(9)}$$

Quando t=0 abbiamo:

```
> eval(subs(t=0,DDG));
```

$$0 \quad \mathbf{(10)}$$

Quindi la funzione G'' si azzera in 0 ed beta, esiste punto 0<omega<beta<alpha<h tale che G'''(omega) = 0.

```
> DDDG := simplify(diff(DDG,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DDDG := \frac{27}{32}\,\mathrm{D}^{(2)}(f)\left(\frac{3}{2}\,t\right) + \frac{27}{32}\,\mathrm{D}^{(2)}(f)\left(-\frac{3}{2}\,t\right) - \frac{27}{32}\,\mathrm{D}^{(2)}(f)\left(-\frac{1}{2}\,t\right) - \frac{27}{32}\,\mathrm{D}^{(2)}(f)\left(\frac{1}{2}\,t\right) + \frac{81}{64}\,t\,\mathrm{D}^{(3)}(f)\left(-\frac{3}{2}\,t\right) + \frac{9}{64}\,t\,\mathrm{D}^{(3)}(f)\left(-\frac{1}{2}\,t\right) - \frac{9}{64}\,t\,\mathrm{D}^{(3)}(f)\left(\frac{1}{2}\,t\right) - \frac{81}{64}\,t\,\mathrm{D}^{(3)}(f)\left(\frac{3}{2}\,t\right) - E\,t^{N-3}\,h^{-N}\,N^3 + 3\,E\,t^{N-3}\,h^{-N}\,N^2 - 2\,E\,t^{N-3}\,h^{-N}\,N \quad \mathbf{(11)}$$

```
> eval(subs(t=0,DDDG));
```

$$0 \quad \mathbf{(12)}$$

Quindi la funzione G''' si azzera in 0 ed beta, esiste punto 0<zeta<omega<beta<alpha<h tale che G''''(zeta) = 0.

```
> DDDDG := simplify(diff(DDDG,t)) assuming N::integer, t>0, h>0, N>2;
```

$$DDDDG := \frac{9}{16}\,\mathrm{D}^{(3)}(f)\left(-\frac{1}{2}\,t\right) - \frac{9}{16}\,\mathrm{D}^{(3)}(f)\left(\frac{1}{2}\,t\right) - \frac{243}{128}\,t\,\mathrm{D}^{(4)}(f)\left(-\frac{3}{2}\,t\right) - \frac{9}{128}\,t\,\mathrm{D}^{(4)}(f)\left(-\frac{1}{2}\,t\right) - \frac{9}{128}\,t\,\mathrm{D}^{(4)}(f)\left(\frac{1}{2}\,t\right) - \frac{243}{128}\,t\,\mathrm{D}^{(4)}(f)\left(\frac{3}{2}\,t\right) - E\,t^{N-4}\,h^{-N}\,N^4 + 6\,E\,t^{N-4}\,h^{-N}\,N^3 - 11\,E\,t^{N-4}\,h^{-N}\,N^2 + 6\,E\,t^{N-4}\,h^{-N}\,N \quad \mathbf{(13)}$$

Ricavo l'errore

```
> ERR := simplify(solve(eval(subs(N=5,t=w,DDDDG)),{E}));
```

$$ERR := \left\{E = \frac{3}{5120}\,\frac{1}{w}\left(\left(8\,\mathrm{D}^{(3)}(f)\left(-\frac{1}{2}\,w\right) - 8\,\mathrm{D}^{(3)}(f)\left(\frac{1}{2}\,w\right) - 27\,w\,\mathrm{D}^{(4)}(f)\left(-\frac{3}{2}\,w\right) - w\,\mathrm{D}^{(4)}(f)\left(-\frac{1}{2}\,w\right) - w\,\mathrm{D}^{(4)}(f)\left(\frac{1}{2}\,w\right) - 27\,w\,\mathrm{D}^{(4)}(f)\left(\frac{3}{2}\,w\right)\right)h^5\right)\right\} \quad \mathbf{(14)}$$

Uso il teorema di Lagrange per semplificare

```
> simplify(subs( ((D@@3)(f))((1/2)*w)=((D@@3)(f))(-(1/2)*w)+w*((D@@4)
  (f))(w1),
          ERR)) ;
```

$$\left\{E = -\frac{3}{5120}\left(8\,D^{(4)}(f)(w1) + 27\,D^{(4)}(f)\left(-\frac{3}{2}w\right) + D^{(4)}(f)\left(-\frac{1}{2}w\right) + D^{(4)}(f)\left(\frac{1}{2}w\right) + 27\,D^{(4)}(f)\left(\frac{3}{2}w\right)\right)h^5\right\} \tag{15}$$

Essendo una somma di derivate quarte posso semplificare con una derivata quarta che prende il valore del valor medio

```
> s := (8+27+1+1+27) / 5120 ;
```

$$s := \frac{1}{80} \tag{16}$$

E quindi l'errore è -3/80 * f''''(zeta)